UR1987_03

1987/03. Modifications to Boonstra and de Ridders' computer programs for numerical modelling of groundwater basins on the Department of Mines Perkin-Elmer minicomputer.

B. D. Weldon

Abstract

The suite of programs presented by Boonstra and de Ridder for numerical modelling of groundwater basins have been modified so as to run on the Department of Mines Perkin-Elmer minicomputer under FORTRAN VII. The modifications require that alterations be made to each source program before calculating a new model. The data file layout has been modified also. This report details the files used, the data file layouts and presents a short discussion on time discretisation as used by the programs.

INTRODUCTION

It is assumed that the reader is familiar with the concepts of numerical modelling of groundwater basins as explained by Boonstra and de Ridder (1981). The modified suite of programs consists of several modules (table 1). Each module comprises several files (table 2). All modules contain dimensioning statements for certain variables. These must be reset if a new model is to be calculated. This is done through the program NEWMOD.CSS.

Table 1. MODULES USED IN THE NUMERICAL ANALYSIS OF GROUNDWATER BASINS

Module	Description	Boonstra and de Ridder
GWMOD1	reading of input data/nodal network	(equivalent to SGMP 1)
GWMOD2	calculation: iteration technique	(equivalent to SGMP 2)
GWMOD3	print-out of results	(equivalent to SGMP 3a)
GWMOD4	plot of results to the line printer	(equivalent to SGMP 3b)

Table 2. FILES THAT ARE USED BY THE VARIOUS MODULES

Module	Source file	Data	Print-out	Binary output	Binary input	CSS
GWMOD1	GWMOD1.FTN	GWMOD1.DAT	GWMOD1.PRT	GWMOD1.BIN		GW1.CSS
GWMOD2	GWMOD2.FTN	GWMOD2.DAT	GWMOD2.PRT	GWMOD2.BIN	GWMOD1.BIN	GW2.CSS
GSMOD3	GWMOD3.FTN	GWMOD3.DAT	GWMOD3.PRT		GWMOD2.BIN	GW3.CSS
GWMOD4	GWMOD4.FTN	GWMOD4.DAT	GWMOD4.PRT		GWMOD2.BIN	GW4.CSS
ALL	ALL	ALL	GWMOD4.PRT	ALL	ALL	GW.CSS

NEWMOD.CSS

All modules contain dimensioning statements for certain variables. These must be reset if a new model is to be calculated. This is done through the program named NEWMOD.CSS. This program calls each module source program into the EDITOR so that new values for:

NN the number of internal nodes and
TNN the total number of nodes

may be amended to suit the new model. For plotting the results, the variable:

IMAJOR, the time discretisation variable MAJOR renamed

must also be amended. In all modules, these changes are made to line number 6, and are made by typing NEWMOD @1, @2 where @1 = NN and @2 = TNN.

The edited source programs are then compiled and linked but are not executed. This must be done using the relevant .CSS program to run each or all modules as required (see table 2).

The following provides a description of each input item required by the various modules. These items are more fully discussed by Boonstra and de Ridder and the reader who is contemplating running a model is strongly advised to become familiar with that discussion.

DATA FILE FOR GWMOD1

In this module, the data of the nodal network, and the data of the hydrogeological parameters and other variables are presented and prepared 80 columns wide.

LINE 1	col 1-80	TITLE, the first 80 characters of the title.
LINE 2	col 1-80	TITLE, continuation of (leave blank otherwise).
LINE 3	col 1-4	NSCONF, the number of internal nodes that represent represent a semi-confined aquifer numbered consecutively, starting with 1 (see Note 1). FORMAT=IIII.
	col 5-8	NCONF, the number of internal nodes that represent a confined aquifer numbered consecutively starting from NSCONF + 1. FORMAT = IIII.
	col 9-12	NEXTN, the number of external nodes numbered consecutively starting from NCONF + 1. FORMAT = IIII.

The total number of nodes TNN = NN + NEXTN.

LINE 4	two columns xNN	NSIDE(K), for each internal node, the number of sides to the polygon forming the internal node. The program places a limit of 7 on the number of sides. FORMAT = II

The actual number of 80 column lines is NN/40

LINE 5	five columns x TNN	CO(I,J), for each node (both internal and external) the x and y coordinates. They are measured in centimetres with regard to an arbitrarily chosen cartesian coordinate system laid on the map of the network configuration (see Note 2). FORMAT = NNDNN.

Actual number of 80 column lines = 2 x TNN/16.

LINE 6	col 1-8	TMBAS, the dimension of time step. It is expressed as a word: MONTH WEEK DAY HOUR. FORMAT = AAAAAAAA.
	col 9-12	DMTIM the number of days in a unit of TMBAS. FORMAT = NNDN.
	col 13-16	SCALE, scale of the map used for the nodal network and from which the coordinates of the nodes are read. FORMAT = NNNNNNND.
	col 17-24	LSW1, an external switch that can be given the value 1 or 2. FORMAT = IIII.
LINE 7	col 1-4	DELTA, the time step. The dimension of DELTA is TMBAS (see Note 4). FORMAT = NDNN.
	col 5-8	MINOR, the number of DELTA periods required to give a unit of TMBAS. FORMAT = III.
	col 9-12	MAJOR, the number of TMBAS's in the total time period considered. FORMAT = III.
	col 13-16	LIST, the number of MAJOR periods in the total time period considered. FORMAT = III.

The total time period in units of TMBAS is the algebraic product DELTA x MINOR x MAJOR x LIST.

LINE 8	col 1-8	ERROR, a tolerance level that directly affects the accuracy of the final results (see Note 5). FORMAT = NNNNNDNN.
	col 9-16	COEFFA, the relaxation coefficient which is in the range 0.8 to 1.2. FORMAT = NNNNNDNN.

LINE 9	five columns x TNN	BL(K), the bottom boundary of the aquifer system, in each nodal area, weighted over the nodal area and at each external node that represents a head-controlled boundary (m above a reference level). Enter zero for external nodes that represent a flow-controlled boundary. FORMAT = NNND.

Actual number of 80 column lines is TNN/20.

LINE 10	col 1-3	K, the number of the internal node. FORMAT = III.
	col 4-10	blank.
	next 3 cols *	NS(J), the number of an adjacent node. FORMAT = III.
	next 5 cols *	PERM(J), the mean horizontal hydraulic conductivity along the side of the aquifer between node K and NS(J), weighted over the length of the side. The dimension is fixed: m/day (see Note 6). FORMAT = NNDNN.
	next 2 cols *	blank.

The sequence indicated by *'s above is repeated for each internal node K. Total number of 80 column lines is NN.

LINE 11	seven columns x NN	CO(I,1), the mean storage coefficient or specific yield of the aquifer in each nodal area, weighted over the nodal area. FORMAT =NDNNNNN.

Total number of 80 column lines is NN/11.

LINE 12	eight columns x NSCONF	PCONF(K), the mean vertical hydraulic conductivity of the confining layer in those nodal areas denoted semi-confined, weighted over the nodal area (m/day). FORMAT = NNNDNNNN.

Total number of 80 column lines is NSCONF/10.
If NSCONF is zero, this line is skipped.

LINE 13	seven columns x NSCONF	ASC(I), the mean specific yield of the confining layer in those nodal areas denoted as semi-confined, weighted over the nodal area. FORMAT = NDNNNNN.

Total number of 80 column lines is NN/11.
If NSCONF is zero, this line is skipped.

LINE 14	four columns	SL(K), the elevation of the land surface in each nodal area, weighted over the nodal area (m above a reference level). FORMAT = NNND.

Total number of 80 column lines is NN/20.

LINE 15	four columns x (NCONF+ NSCONF)	CO(K,2), the thickness of the aquifer if the nodes are denoted as semi-confined or confined, weighted over the nodal area. FORMAT = NNND.

Total number of 80 column lines is (NSCONF + NCONF)/20.
If both NSCONF and NCONF are equal to zero, this line is skipped.

LINE 16 col 1-4 LSW2, an external switch that can be assigned the values 1 or 2 (see Note 7). FORMAT = IIII.

col 5-8 DELQ, a coefficient that determines the speed of reaching the final value of the additional percolation or abstraction rate that will keep the water table elevations between the prescribed limits (see Note 8). FORMAT = NNDN.

LINE 17 eight columns x NN UL(K), the upper limit to which the water table in each nodal area is allowed to rise (m above a reference level). FORMAT = NNNDNNNN.

Total number of 80 column lines is NN/10.
If LSW2 is equal to 2, this line is skipped.

LINE 18 eight columns x NN OL(K), the lower limit to which the water table in each nodal area is allowed to fall (m above a reference level). FORMAT = NNNDNNNN.

Total number of 80 column lines is NN/10.
If LSW2 is equal to 2, this line is skipped.

Note 1. FORMAT denotes the number of characters and their type in the field allowed for the named variable thus:

I = numeric character of an integer number
N = numeric character of a real number
D = decimal point (.)
A = alphabetic character

Note 2. Where the triangle formed by three nodes possess an angle greater than 90 degrees a message is printed indicating the nodes concerned.

Note 3. The unit of length is always fixed at one metre; it is used in the hydraulic conductivity values (metres/day) and in the elevation values (metres above a reference datum). LSW1 is used to control the size of the nodal areas and when set to one it gives a unit of area in $metres^2$; when set to two it gives a unit of area of one million $metres^2$.

Note 4. It is advisable to run the program with various DELTA values. When the results are compared, it can readily be seen for what value of DELTA the results do not appreciably change.

Note 5. If after 50 iterations, ERROR is still exceeded, the iteration is terminated and a message 'Relaxation fails to converge' is printed. Calculation resumes for the next time step.

Note 6 When the values specified for the hydraulic conductivity on the common side of two internal nodes differs, a message to this affect is printed.

Note 7. Setting LSW2=1 means that the water table elevations are prescribed within certain limits and that specific input values (UL(K) and OL(K)) must be given for some or all of the nodal areas.

Note 8. Determining the value of DELQ is a matter of trial and error. A value of 0.1 has been found to work satisfactorily.

TIME DISCRETISATION

In modelling a groundwater basin two time parameters are indispensable. The first is the time step (DELTA), which is implicit to the finite difference method. Common time steps are a week, fortnight, or a month for regional studies and a day for local studies like aquifer tests. The second (LIST) is the total time period for which the calculations are to be made.

Two other time parameters (MINOR and MAJOR) are also used in the program suite. They are included to allow the boundary conditions to be prescribed at time intervals that are different from the time step (DELTA) e.g. each week or month depending on the available data; and to provide print-out of the water table elevations at periods of time other than the time step e.g. print-out for each month giving the water balance components for each year even though the actual time step is a week.

The relationships between the different time units is best illustrated in the following examples:

For regional studies -

Parameter	Value	Interpretation
TMBAS	MONTH	month
DELTA	0.5	fortnight
MINOR	2	number of fortnights in a month
MAJOR	12	number of months in a year
LIST	5	number of years in total time period

In the above context 1 year = 12 months = 24 fortnights = 48 weeks = 336 days.

For local studies -

Parameter	Value	Interpretation
TMBAS	HOURS	hour
DELTA	0.1	6 minutes
MINOR	10	number of 6 min. periods in an hour
MAJOR	24	number of hours in a day
LIST	2	number of days in total time period

Total time (in units of TMBAS) = DELTA x MINOR x MAJOR x LIST

DATA FILE FOR GWMOD2

In this module, the water table elevations at the nodes and the water balance of the nodal areas are calculated for each time step. The results are stored in a binary output file.

LINE 1 col 1-4 — LSW3, an external switch that can be given the values 1, 2, or 3. It determines the frequency with which the boundary conditions must be specified (see Note 1). FORMAT = IIII.

col 5-8 — LSW4, an external switch that can be prescribed the values 1 or 2. It determines the frequency with which the boundary conditions are read (see Note 2). FORMAT = IIII.

col 9-12 — LSW5, an external switch that can be given the values 1 or 2. It determines whether the time step used in the calculation is fixed or variable (see Note 3). FORMAT = IIII.

col 13-16 — LSW6, an external switch that can be assigned the values 1 or 2. It determines whether the water table elevation in the top layer is variable (for semi-confined aquifers) or fixed (see Note 4). FORMAT = IIII.

LINE 2 eight columns x TNN — H(K), the initial water table elevations of the aquifer for all internal and external nodes, measured at the nodes themselves (m above a reference level). FORMAT = NNNNNDNN.

Total number of 80 column lines is TNN/10.

LINE 3 eight columns x NSCONF — HCONF(K), the initial water table elevations of the top layer for those internal nodes that are denoted as semi-confined (m above a reference level). FORMAT = NNNNNDNN.

Total number of 80 column lines is NSCONF/10.
If NSCONF equals zero, this line is skipped.

LINE 4 col 1-8 reserved for identification code.

next 5 columns x NN — RECH(K), the net recharge in each internal nodal area being the sum of a number of external flows with dimension depth per time. The unit length is fixed at one metre and the unit time is TMBAS. A positive sign means recharge to the aquifer, or if the aquifer is semi-confined, recharge to the confining layer (see Note 5).
FORMAT = NDNNN.

Total number of 80 column lines is NN/14.

LINE 5 col 1-8 reserved for identification code.

next 10 columns x NN — FLWCON(K), net recharge rate in each internal nodal area being the sum of a number of external flows with dimension volume per time. The unit of volume is determined by LSW1 (1 m^3 or 1 million m^3); the unit of time is TMBAS. A positive sign means recharge to the aquifer (see Note 5).
FORMAT = NNNNNDNNNN.

Total number of 80 column lines is NN/7.

LINE 6 col 1-8 reserved for identification code.

next 8 columns x NEXTN — H(K), water table elevations in the external nodes that simulate head-controlled boundaries (m above a reference level). FORMAT = NNNNNDNN

Total number of 80 column lines is NEXTN/9.

LINE 7 col 1-6 DELTA, the time step. This is required if LSW5 equals 2. FORMAT = NDNNNN.

If LSW5 is equal to one, this line is skipped.

Note 1. Lines 4 to 6 are repeated:

LIST times if LSW3=1
LIST x MAJOR times if LSW3=2
LIST x MAJOR x MINOR times if LSW3=3.

Note 2. LSW4 = 2 during calibration i.e. the boundary conditions are read for each DELTA, first or second acculmulation level depending on the value of LSW3; in operational runs LSW4 is usually set to 1.

Note 3. LSW5 = 2 indicates a variable time step. This is often used in simulating aquifer tests (logarithmic time step) but for regional groundwater flow problems a fixed time step is preferred.

Note 4. LSW6 = 2 is valid when the suite of programs is used to simulate aquifer tests as it is then assumed that the water table elevation in the top layer remains constant during the test. For regional groundwater flow problems, LSW6 must be given the value one.

Note 5. For unconfined aquifers both RECH(K) and FLWCON(K) can be used; for confined aquifers only FLWCON(K) can be used and for semi-confined aquifers RECH(K) can be used for the confining layer and FLWCON(K) for the aquifer itself.

DATA FILE FOR GWMOD3

This module presents the results of the calculations in tabular form. The input data comprises a binary file created by GWMOD2 and the values of three external switches. The data file is 80 columns wide.

LINE 1 col 1-4		LSW7, an external switch that can be given the values 1, 2 or 3. It determines the frequency with which the water table elevations are printed. FORMAT = IIII.
col 5-8		LSW8, an external switch that can be prescribed the values 1, 2 or 3. It determines the frequency with which the water balance of the nodal area is calculated. The results are printed after each particular time level. FORMAT = IIII.
col 9-12		LSW9, an external switch which may take the values 1 or 2. It determines whether the water table levels of the internal nodes only or all nodes are printed (see Note 1). FORMAT = IIII.

Note 1. LSW9 = 2 is usually done only once to check the head-controlled boundaries numerically.

DATA FILE FOR GWMOD4

This module presents the water table elevations at the nodes in graphical form on a line printer. The binary file generated by GWMOD2 is used as a data source along with various characters or symbols for the line printer plots. In addition, the value of three external switches are supplied. For the calibration process, the historical water table elevations are prescribed.

LINE 1 col 1-4		these columns are always left blank.
col 5-8		HI, a character or symbol to denote the unit of scale. FORMAT = AAAA.
col 9-12		HMINUS, a character or symbol that denotes the tenth of the unit of scale. FORMAT = AAAA.
col 13-16		HX, a character or symbol that indicates when a computed water table elevation falls outside the maximum range of the scale. FORMAT = AAAA.
col 17-20		ASTRSK, a character or symbol that indicates the position of a calculated water table elevation in the graph. FORMAT = AAAA.
col 21-24		HY, a character or symbol that indicates when a

		historical water table elevation falls outside the maximum range of the scale. FORMAT = AAAA.
	col 25-28	PLUS, a character or symbol that indicates the position of a historical water table elevation in the graph. FORMAT = AAAA.
	col 29-30	LSW9, an external switch that may take the values 1 or 2. It determines whether the water table elevations are plotted for the internal nodes only or for all nodes. FORMAT = II.
	col 31-32	LSW10, an external switch that may be prescribed the values 1 or 2. It determines whether the historical and calculated water table elevations are plotted together or only the calculated values are plotted. FORMAT = II.
	col 33-34	LSW11, an external switch that may be assigned the values 1 or 2. It determines whether the deviation between the calculated and historical water table elevations is calculated or not. FORMAT = II.
LINE 2	col 1-4	MINOR, as previously defined (module GWMOD1). FORMAT = IIII.
	col 5-8	MAJOR, as previously defined (module GWMOD1). FORMAT = IIII.
	col 9-12	LIST, as previously defined (module GWMOD1). FORMAT = IIII.

These parameters may be given new values here so as to reduce the amount of data generated when making operational runs over long time periods.
If LSW10 is equal to 1, this line is skipped.

LINE 3	col 1-8	reserved for identification code.
	eight columns x NN	H(K,M), the historical water table elevation of the aquifer at each internal node. FORMAT = NNNNNDNN.

Total number of 80 column lines is NN/9.
If LSW10 is equal to 2, this line is skipped.
Line 3 is repeated LIST x MAJOR times.

Note 1. LSW9 = 2 is usually done only once to provide a visual check of the head controlled boundaries.

Note 2. LSW10 = 1 is required for the calibration process; in operational runs there is no historical water table elevations.

Note 3. LSW11 = 2 is required for operational runs where there is no historical water table elevations.

ACKNOWLEDGEMENTS

P. C. Stevenson kindly typed the program listings printed in Boonstra and de Ridder and commenced modifications for the FORTRAN version running on the Department of Mines Perkin-Elmer minicomputer. A. L. Telfer guided the drafting of the .CSS's and assisted in de-bugging of the programs.

REFERENCES

BOONSTRA, J.; DE RIDDER, N. A. 1981. Numerical modelling of groundwater basins. *Publ. int. Inst. Land Reclam. Improv.* 29.

[23 January 1987]

APPENDIX A
CSS and FORTRAN program listings

```
*NEWMOD.CSS a program to modify the groundwater modules
* PARAMETERS: @1=NN, @2=TNN
$IFNUL @1; $WR NN and TNN must both be specified; $EXIT; $ENDC
$IFNUL @2; $WR NN and TNN must both be specified; $EXIT; $ENDC
$BUILD GW.CMD
OP INPLACE=ON
G GWMOD1.FTN
COL17,/(NN=@1,TNN=@2)     /,6
G GWMOD2.FTN
COL17,/(NN=@1,TNN=@2)     /,6
G GWMOD3.FTN
COL17,/(NN=@1,TNN=@2)     /,6
G GWMOD4.FTN
COL17,/(NN=@1,TNN=@2)     /,6
END
$ENDB
L EDIT32/S
ST, LIST=CON:,COM=GW.CMD
XDE GW.CMD
SYSPRT NULL:
COMPLINK GWMOD1.FTN
COMPLINK GWMOD2.FTN
COMPLINK GWMOD3.FTN
COMPLINK GWMOD4.FTN
SYSPRT CON:
DE GWMOD1.OBJ
DE GWMOD2.OBJ
DE GWMOD3.OBJ
DE GWMOD4.OBJ
$EXIT
```

```
*GW.CSS  TO RUN THE ENTIRE PROGRAMME SUITE
*        OF INPUT,CALCULATION,PRINTING & PLOTTING MODULES
DE GWMOD1.PRT,GWMOD1.BIN
RUN GWMOD1.FTN
PRI GWMOD1.PRT,DEV=EPR:
DE GWMOD2.PRT,GWMOD2.BIN
RUN GWMOD2.FTN
PRI GWMOD2.PRT,DEV=EPR:
DE GWMOD3.PRT
RUN GWMOD3.FTN
PRI GWMOD3.PRT,DEV=EPR:
DE GWMOD4.PRT
RUN GWMOD4.FTN
PRI GWMOD4.PRT,DEV=EPR:
$EXIT
```

```
*GW1.CSS  TO RUN DATA INPUT MODULE GWMOD1.FTN
DE GWMOD1.PRT
DE GWMOD1.BIN
RUN GWMOD1.FTN
PRI GWMOD1.PRT,DEV=EPR:
$EXIT


$TITL GWMOD1.FTN
C      GROUNDWATER MODEL PART1: READING INPUT DATA(POLYGONS/INHOMOGENEOUS)
       INTEGER TNN
C
C
C
       PARAMETER (NN=5,TNN=15)
C
       REAL LENGTH
       DIMENSION PERM(7),NS(7),TITLE(40),TMBAS(2)
       DIMENSION ASC( NN),UL( NN),OL( NN),AREA( NN),AS( NN),SL( NN)
       DIMENSION CO(TNN,2)
       DIMENSION BL(TNN),NSIDE( NN),PCONF( NN),NREL( NN,7),CONDU( NN,7)
C
       OPEN(UNIT=8,FILE='GWMOD1.DAT',STATUS='OLD')
       OPEN(UNIT=5,FILE='GWMOD1.PRT',STATUS='NEW')
       OPEN(UNIT=1,FILE='GWMOD1.BIN',STATUS='NEW')
C
       READ(8,40) TITLE
40     FORMAT(20A4/20A4)
       READ(8,1) NSCONF,NCONF,NEXTN
1      FORMAT(3I4)
       NO=NN+1
       NCONF=NCONF+NSCONF
       NPH=NN-NCONF
       NNC=NCONF-NSCONF
       DO 47 I=1,NN
       DO 47 J=1,7
       NREL(I,J)=0
47     CONDU(I,J)=0.0
       READ(8,10) (NSIDE(K),K=1,NN)
10     FORMAT(40I2)
       READ(8,42) ((CO(I,J),J=1,2),I=1,TNN)
42     FORMAT(16F5.2)
       READ(8,46) TMBAS,DMTIM,T,SCALE,LSW1
46     FORMAT(2A4,2F4.1,F8.0,I4)
       READ(8,3) DELTA,MINOR,MAJOR,LIST
3      FORMAT(F4.2,3I4)
       READ (8,4)ERROR,COEFFA
4      FORMAT(2F8.2)
       WRITE(5,44)TITLE
44     FORMAT('GROUNDWATER MODEL FOR ',20A4/20A4//)
       WRITE(5,12)NN,NEXTN,NSCONF,NNC,NPH
12     FORMAT(' NUMBER OF INTERNAL NODES IS',I4/' NUMBER OF EXTERNAL N'
      1'ODES IS',I4///' NUMBER OF SEMICONFINED NODAL AREAS IS',I4/' N'
      2'UMBER OF CONFINED NODAL AREAS      IS',I4/' NUMBER OF UNCONFIN'
      3'ED NODAL AREAS   IS',I4/)
       IF(LSW1.EQ.1) WRITE(5,17)
       IF(LSW1.EQ.2) WRITE(5,39)
17     FORMAT(//' UNIT LENGTH :          1 METRE'/' UNIT AREA   : 1,000'
```

```
     1',000 SQ.METRES'/' UNIT VOLUME : 1,000,000 CU.METRES')
39    FORMAT(//' UNIT LENGTH : 1 METRE'/' UNIT AREA : 1 SQ.METRE'/
     1' UNIT VOLUME : 1 CU. METRE')
      WRITE(5,45) TMBAS
45    FORMAT(//' UNIT OF TIME FOR DELTA AND BOUNDARY CONDITIONS IS '
     3,2A4)
      WRITE(5,13)
13    FORMAT(//13H BETWEEN NODE,4X,12HPERMEABILITY,4X,5HWIDTH,4X,6HLENGT
     1H,18X,8HNODE NO.,10X,8HX-COORD.,10X,8HY-COORD.,10X,6HBOTTOM/18X,10
     2H   M/DAY  ,6X,2HCM,8X,2HCM,41X,2HCM,15X,2HCM,13X,9HELEVATION//)
      READ(8,6)(BL(K),K=1,TNN)
      TAREA=0.
      DO 11 I=1,NN
      J2=NSIDE(I)
      READ(8,15) K,(NS(J),PERM(J),J=1,J2)
15    FORMAT(I3,7X,7(I3,F5.2,2X))
      DO 41 J=1,J2
      NREL(K,J)=NS(J)
41    CONDU(K,J)=PERM(J)
11    CONTINUE
      DO 49 K=1,NN
      J2=NSIDE(K)
      DO 52 J=1,J2
      NOD=NREL(K,J)
      IF(NOD.LT.K.OR.NOD.GT.NN) GO TO 52
      J3=NSIDE(NOD)
      DO 54 J1=1,J3
      IF(NREL(NOD,J1).NE.K) GO TO 54
      GO TO 55
54    CONTINUE
55    IF(CONDU(K,J).EQ.CONDU(NOD,J1)) GO TO 52
      WRITE(5,53) K,NOD,NOD,K
53    FORMAT(30H PERMEABILITIES BETWEEN NODES ,I3,1H/,I3,5H AND ,I3,1H/,
     1I3,17H ARE NOT THE SAME)
52    CONTINUE
49    CONTINUE
      JJ=0
      DO 18 K=1,NN
      J2=NSIDE(K)
      DUM=0.0
      DO 35 J=1,J2
      JJ=JJ+1
      NS(1)=NREL(K,J)
      IF(J-1.EQ.0) NS(2)=NREL(K,J2)
      IF(J-1.NE.0) NS(2)=NREL(K,J-1)
      IF(J.EQ.J2) NS(3)=NREL(K,1)
      IF(J.NE.J2) NS(3)=NREL(K,J+1)
      DO 36 I=1,3
      X=CO(K,1)-CO(NS(I),1)
      Y=CO(K,2)-CO(NS(I),2)
36    PERM(I)=X*X+Y*Y
      DO 37 I=2,3
      X=CO(NS(1),1)-CO(NS(I),1)
      Y=CO(NS(1),2)-CO(NS(I),2)
37    PERM(I+2)=X*X+Y*Y
      B2=PERM(2)+PERM(4)-PERM(1)
      IF(B2.LT.0.0) WRITE(5,48) K,NS(2),NS(1)
      B2=PERM(3)+PERM(5)-PERM(1)
```

```
      IF(B2.LT.0.0) WRITE(5,48) K,NS(3),NS(1)
48    FORMAT(/30H THE ANGLE FORMED BY THE NODES,I3,2H ,,I3,2H ,,I3,27H I
     1S GREATER THAN 90 DEGREES/)
      A=PERM(1)
      B=(PERM(4)+PERM(2)-PERM(1))**2
      C=(PERM(3)+PERM(5)-PERM(1))**2.
      D=4*PERM(4)*PERM(2)
      E=4*PERM(3)*PERM(5)
      WIDTH=SQRT((0.25*A*B)/(D-B))+SQRT((0.25*A*C)/(E-C))
      LENGTH=SQRT(A)
      IF(JJ.GT.TNN) WRITE(5,16) K,NS(1),CONDU(K,J),WIDTH,LENGTH
      IF(JJ.LE.TNN) WRITE(5,16) K,NS(1),CONDU(K,J),WIDTH,LENGTH,JJ,CO(
     1JJ,1),CO(JJ,2),BL(JJ)
16    FORMAT(I5,4H AND,I4,10X,F5.2,5X,F5.2,5X,F5.2,18X,I5,15X,F5.2,13X,F
     15.2,F16.0)
      IF(LSW1.EQ.1)CONDU(K,J)=CONDU(K,J)*(WIDTH/LENGTH)*DMTIM/1000000.
      IF(LSW1.EQ.2)CONDU(K,J)=CONDU(K,J)*(WIDTH/LENGTH)*DMTIM
      DUM=DUM+WIDTH*LENGTH
35    CONTINUE
      IF(LSW1.EQ.1) AREA(K)=DUM*0.25*(SCALE/100000.)**2
      IF(LSW1.EQ.2) AREA(K)=DUM*0.25*(SCALE/100.)**2
18    TAREA=TAREA+AREA(K)
      READ(8,43)(CO(I,1),I=1,NN)
43    FORMAT(11F7.5)
      IF(NSCONF.GT.0) READ(8,5) (PCONF(K),K=1,NSCONF)
5     FORMAT(10F8.4)
      IF(NSCONF.GT.0) READ(8,43) (ASC(I),I=1,NSCONF)
      DO 38 K=1,NN
38    AS(K)=AREA(K)*CO(K,1)
      READ(8,6)(SL(K),K=1,NN)
      IF(NCONF.NE.0) READ(8,6) (CO(I,2),I=1,NCONF)
6     FORMAT(20F4.0)
      READ(8,2) LSW2,DELQ
2     FORMAT(I4,F4.1)
      IF(LSW2.EQ.2) GO TO 7
      READ(8,5) (UL(K),K=1,NN)
      READ(8,5)(OL(K),K=1,NN)
      GO TO 9
7     DO 8 K=1,NN
      UL(K)=SL(K)
8     OL(K)=BL(K)
9     WRITE (5,19)
19    FORMAT(//5H NODE,5X,4HAREA,5X,14HSTORAGE COEFF.,6X,7HSURFACE,8X,9H
     1THICKNESS,7X,9HTHICKNESS,5X,12HPERMEABILITY,5X,14HSTORAGE COEFF./2
     22X,7HAQUIFER,9X,9HELEVATION,5X,12HCONF.AQUIFER,6X,8HTOPLAYER,7X,8H
     3TOPLAYER,10X,8HTOPLAYER/)
      DUM=0.0
      DO 50 K=1,NN
      DL=BL(K)+CO(K,2)
      IF(K.LE.NSCONF) WRITE(5,20) K,AREA(K),CO(K,1),SL(K),CO(K,2),DL,P
     1CONF(K),ASC(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,20) K,AREA(K),CO(K,1),SL(
     1K),CO(K,2)
      IF(K.GT.NCONF) WRITE(5,20) K,AREA(K),CO(K,1),SL(K)
20    FORMAT(I3,F11.2,F14.5,F17.0,F15.0,F16.0,F16.3,F18.3)
50    CONTINUE
      WRITE(5,33) TAREA
33    FORMAT(/F14.2)
```

```
      IF(NSCONF.EQ.0) GO TO 34
      DO 51 K=1,NSCONF
      PCONF(K)=PCONF(K)*DMTIM
51    ASC(K)=AREA(K)*ASC(K)
34    IF(LSW2.EQ.2) GO TO 23
      WRITE(5,21)
21    FORMAT(//5H NODE,5X,12HUPPER LIMITS,5X,12HLOWER LIMITS//)
      WRITE(5,22) (K,UL(K),OL(K),K=1,NN)
22    FORMAT(I3,F16.2,F17.2)
23    WRITE(5,24)
24    FORMAT(//4X,23HMISCELLANEOUS CONSTANTS,10X,27HVALUES OF EXTERNAL S
     1WITCHES/)
      WRITE (5,25)LIST,LSW1
25    FORMAT(6X,7HLIST = ,I7,26X,7HLSW1 = ,I2)
      WRITE(5,31) MAJOR,LSW2
31    FORMAT(5X,8HMAJOR = ,I7,26X,7HLSW2 = ,I2)
      WRITE(5,32) MINOR
32    FORMAT(5X,8HMINOR = ,I7)
      WRITE(5,27)SCALE
27    FORMAT(5X,8HSCALE = ,F10.2)
      WRITE(5,28) COEFFA
28    FORMAT(4X,9HCOEFFA = ,F10.2)
      WRITE(5,14) DMTIM
14    FORMAT(5X,8HDMTIM = ,F10.2)
      WRITE(5,26) DELTA
26    FORMAT(5X,8HDELTA = ,F14.6)
      WRITE(5,29)ERROR
29    FORMAT(5X,8HERROR = ,F14.6)
      WRITE(5,30)DELQ
30    FORMAT(6X,7HDELQ = ,F14.6)
      WRITE(1) DELQ,COEFFA,T,NO,ERROR,DELTA,NCONF,LSW1
      WRITE(1) TITLE,TMBAS,(BL(K),K=1,TNN),LIST,MAJOR,MINOR,NSCONF
      WRITE(1) (UL(K),OL(K),SL(K),AREA(K),AS(K),NSIDE(K),K=1,NN)
      WRITE(1) ((NREL(K,J),CONDU(K,J),J=1,7),K=1,NN)
      IF(NCONF.GT.0) WRITE(1)(CO(I,2),I=1,NCONF)
      IF(NSCONF.GT.0) WRITE(1) (ASC(K),PCONF(K),K=1,NSCONF)
      END
```

```
*GW2.CSS  TO RUN CALCULATION MODULE GWMOD2.FTN
DE GWMOD2.PRT;DE GWMOD2.BIN
RUN GWMOD2.FTN
PRI GWMOD2.PRT,DEV=EPR:
$EXIT


$TITL GWMOD2.FTN
C GROUNDWATER MODEL PART 2 CALCULATION OF GROUNDWATER FLOW
C          GAUSS/SEIDEL
      INTEGER TNN,SUBITR
      REAL IMP
C
      PARAMETER (NN=5,TNN=15)
C
      DIMENSION NREL( NN,7),NSIDE( NN),AREA( NN),TITLE(40),TMBAS(2)
      DIMENSION H(TNN),HO( NN),HT( NN),HCONF( NN),HOCONF( NN),DRECH( NN)
      DIMENSION RECH( NN),FLWCON( NN),TOTALQ( NN),S( NN),QSEEP( NN)
      DIMENSION CONDU( NN,7),PCONF( NN),SC( NN),ASC( NN),AS( NN)
      DIMENSION BL(TNN),UL( NN),OL( NN),THID( NN),SL( NN)
C
      OPEN(UNIT=8,FILE='GWMOD2.DAT',STATUS='OLD')
      OPEN(UNIT=5,FILE='GWMOD2.PRT',STATUS='NEW')
      OPEN(UNIT=1,FILE='GWMOD1.BIN',STATUS='OLD')
      OPEN(UNIT=2,FILE='GWMOD2.BIN',STATUS='NEW')
C
      READ(1) DELQ,COEFFA,T,NO,ERROR,DELTA,NCONF,LSW1
      READ(1) TITLE,TMBAS,(BL(K),K=1,TNN),LIST,MAJOR,MINOR,NSCONF
      READ(1) (UL(K),OL(K),SL(K),AREA(K),AS(K),NSIDE(K),K=1,NN)
      READ(1) ((NREL(K,J),CONDU(K,J),J=1,7),K=1,NN)
      IF(NCONF.GT.0) READ(1) (THID(I),I=1,NCONF)
      IF(NSCONF.GT.0) READ(1) (ASC(K),PCONF(K),K=1,NSCONF)
      READ(8,38) LSW3,LSW4,LSW5,LSW6
38    FORMAT(4I4)
      READ(8,18) (H(K),K=1,TNN)
      IF(NSCONF.GT.0) READ(8,18) (HCONF(K),K=1,NSCONF)
18    FORMAT(10F8.2)
      WRITE(2) LIST,MAJOR,MINOR,DELTA,T,LSW1
      WRITE(2) (H(K),K=1,TNN),NSCONF,(SL(K),K=1,NN),NCONF,TITLE,TMBAS
      IF(NSCONF.GT.0) WRITE(2) (HCONF(K),K=1,NSCONF)
      WRITE(5,40)
40    FORMAT(12H CALCULATION////)
      WRITE(5,39) TITLE
39    FORMAT('GROUNDWATER MODEL FOR ',20A4/20A4)
      WRITE(5,41)
41    FORMAT(///28H VALUES OF EXTERNAL SWITCHES/)
      WRITE(5,42) LSW3,LSW4,LSW5,LSW6
42    FORMAT(10X,'LSW3= ',I2/10X,'LSW4= ',I2/10X,'LSW5= ',I2/10X,
     1'LSW6= ',I2)
      DO 90 L=1,LIST
      IF(LSW3.NE.1)GO TO 31
      READ(8,1) (RECH(K),K=1,NN)
1     FORMAT(8X,14F5.3)
      READ(8,2) (FLWCON(K),K=1,NN)
2     FORMAT(8X,7F10.4)
      DO 32 K=1,NN
32    RECH(K)=RECH(K)*AREA(K)
      READ(8,4) (H(K),K=NO,TNN)
```

```
4       FORMAT(8X,9F8.2)
31      DO 80 M=1,MAJOR
        IF(LSW4.EQ.1.AND.L.NE.1) GO TO 5
        IF(LSW3.NE.2) GO TO 5
        READ(8,1) (RECH(K),K=1,NN)
        READ(8,2) (FLWCON(K),K=1,NN)
        DO 3 K=1,NN
3       RECH(K)=RECH(K)*AREA(K)
        READ(8,4) (H(K),K=NO,TNN)
5       DO 70 JT=1,MINOR
        IF(LSW3.NE.3)GO TO 7
        READ(8,1) (RECH(K),K=1,NN)
        READ(8,2) (FLWCON(K),K=1,NN)
        DO 6 K=1,NN
6       RECH(K)=RECH(K)*AREA(K)
        READ(8,4) (H(K),K=NO,TNN)
7       IF(LSW5.EQ.2) READ(8,28) DELTA
28      FORMAT(F6.4)
        T=T+DELTA
        ITRNO=0
        DO 8 K=1,NN
        DRECH(K)=0.0
8       HO(K)=H(K)
        DO 35 K=1,NSCONF
        QSEEP(K)=0.0
        SC(K)=0.0
35      HOCONF(K)=HCONF(K)
9       ITRNO=ITRNO+1
        SUBITR=0
10      LM=0
        ITOET=1
        SUM=0.0
        DO 14 K=1,NN
        RES=0.0
        IMP=0.0
        HL=H(K)
        J2=NSIDE(K)
        DO 13 J=1,J2
        NOD=NREL(K,J)
        B=(BL(NOD)+BL(K))/2.
        HA=(H(NOD)+HL)/2.
        THI=HA-B
        IF(K.LE.NCONF.AND.NOD.LE.NCONF) THI=(THID(K)+THID(NOD))*0.5
        IF(K.LE.NCONF) THI=THID(K)
        IF(NOD.LE.NCONF) THI=THID(NOD)
        Y=THI*CONDU(K,J)
        IMP=IMP+Y
        Q=Y*(H(NOD)-HL)
        RES=RES+Q
13      CONTINUE
        TOTALQ(K)=RES
        S(K)=(HO(K)-HL)*AS(K)/DELTA
        IF(K.GT.NSCONF) GO TO 33
        DL=BL(K)+THID(K)
        IF(HCONF(K).LE.DL) GO TO 30
        C=(HCONF(K)-DL)/PCONF(K)
        QSEEP(K)=((HCONF(K)-HL)/C)*AREA(K)
        IF(HL.LE.DL) QSEEP(K)=((HCONF(K)-DL)/C)*AREA(K)
```

```
      GO TO 43
30    QSEEP(K)=0.0
43    RES=RES+FLWCON(K)+QSEEP(K)+S(K)+DRECH(K)
      SC(K)=(HOCONF(K)-HCONF(K))*ASC(K)/DELTA
      RESC=RECH(K)+SC(K)-QSEEP(K)
      IF(LSW6.EQ.1) HCONF(K)=HCONF(K)+RESC/(ASC(K)/DELTA+AREA(K)/C)
      H(K)=HL+(RES*COEFFA/(IMP+AS(K)/DELTA+AREA(K)/C))
      GO TO 34
33    IF(K.LE.NCONF) RECH(K)=0.0
      RES=RES+RECH(K)+S(K)+FLWCON(K)+DRECH(K)
      H(K)=HL+(RES*COEFFA/(IMP+AS(K)/DELTA))
34    SUM=SUM+ABS(RES)
14    CONTINUE
      IF(SUM.GT.ERROR) ITOET=2
      IF(SUBITR.LT.49) GO TO 17
      WRITE(5,16)T
16    FORMAT(//26H DELQ IS TOO SMALL AT TIME,F8.4//)
17    IF(ITRNO.LT.50) GO TO 20
      WRITE(5,19) T
19    FORMAT(//60X,37H RELAXATION FAILS TO CONVERGE AT TIME,F8.4)
20    DO 25 K=1,NN
      IF(H(K).GE.OL(K).AND.H(K).LT.UL(K)) GO TO 25
      IF (ABS(FLWCON(K)).GT.ABS(RECH(K))) GO TO 21
      TDRECH=ABS(RECH(K))*DELQ
      GO TO 22
21    TDRECH=ABS(FLWCON(K))*DELQ
22    IF(H(K).LT.OL(K)) GO TO 23
      IF(K.LE.NCONF) GO TO 25
      DRECH(K)=DRECH(K)-TDRECH
      GO TO 24
23    DRECH(K)=DRECH(K)+TDRECH
24    LM=1
25    CONTINUE
      IF(ITRNO.GE.50.OR.SUBITR.GE.50) GO TO 37
      IF(LM.EQ.1) SUBITR=SUBITR+1
      IF(LM.EQ.1) GO TO 10
      IF(ITOET.EQ.2) GO TO 9
37    DO 26 K=1,NN
      IF(K.GT.NSCONF) GO TO 27
      IF(HCONF(K).GT.SL(K)) WRITE(5,29)K,T
29    FORMAT(// 'WATER TABLE IN CONFINING LAYER AT NODE ',I3, 'IS
     1ABOVE SURFACE LEVEL AT TIME ',F8.4)
27    IF(K.GT.NCONF) GO TO 26
      IF(H(K).LT.(BL(K)+THID(K))) WRITE(5,36) K,T
36    FORMAT(//37H PIEZOMETRIC LEVEL IN AQUIFER AT NODE,I3,26H IS BELOW
     1TOPLAYER AT TIME,F8.4)
26    CONTINUE
      WRITE(2) (RECH(K),TOTALQ(K),S(K),FLWCON(K),DRECH(K),K=1,NN)
      WRITE(2) (H(K),K=1,TNN),T,ITRNO,DELTA,SUBITR
      IF(NSCONF.GT.0) WRITE(2) (HCONF(K),QSEEP(K),SC(K),K=1,NSCONF)
70    CONTINUE
80    CONTINUE
90    CONTINUE
      END
```

```
*GW3.CSS TO RUN THE PRINTING MODULE GWMOD3.FTN
DE GWMOD3.PRT
RUN GWMOD3.FTN
PRI GWMOD3.PRT,DEV=EPR:
$EXIT


$TITL GWMOD3.FTN
C  GROUNDWATER MODEL PART 3A ; PRINTING OF RESULTS
      INTEGER TNN,SUBITR
C
C
C
      PARAMETER (NN=5,TNN=15)
C
      DIMENSION H(TNN), SL( NN),HCONF( NN),TITLE(40),TMBAS(2)
      DIMENSION RECH( NN),FLWCON( NN),TOTALQ( NN),S( NN),QSEEP( NN)
      DIMENSION SC( NN),DRECH( NN),TOTRE( NN),TOTFC( NN),TOTSQ( NN)
      DIMENSION TOTS( NN),TOTQS( NN),TOTSC( NN),TOTDQ( NN)
C
      OPEN(UNIT=8,FILE='GWMOD3.DAT',STATUS='OLD')
      OPEN(UNIT=5,FILE='GWMOD3.PRT',STATUS='NEW')
      OPEN(UNIT=2,FILE='GWMOD2.BIN',STATUS='OLD')
C
      READ(2) LIST,MAJOR,MINOR,DELTA,T,LSW1
      READ(2) (H(K),K=1,TNN),NSCONF,(SL(K),K=1,NN),NCONF,TITLE,TMBAS
      IF(NSCONF.GT.0) READ(2) (HCONF(K),K=1,NSCONF)
      READ(8,14) LSW7,LSW8,LSW9
 14   FORMAT(3I4)
      IF(LSW9.EQ.1) NT=NN
      IF(LSW9.EQ.2) NT=TNN
      NEXTN=TNN-NN
      NPH=NN-NCONF
      NNC=NCONF-NSCONF
      WRITE (5,1)
 1    FORMAT (9H SOLUTION////)
      WRITE(5,44) TITLE
 44   FORMAT(' GROUNDWATER MODEL FOR ',20A4/20A4//)
      WRITE(5,43) NN,NEXTN,NSCONF,NNC,NPH
 43   FORMAT(' NUMBER OF INTERNAL NODES IS',I4/' NUMBER OF EXTERNAL'
     1' NODES IS',I4///' NUMBER OF SEMICONFINED NODAL AREAS IS',I4/
     2' NUMBER OF CONFINED NODAL AREAS      IS',I4/' NUMBER OF '
     3'UNCONFINED NODAL AREAS   IS',I4/)
      IF(LSW1.EQ.1) WRITE(5,42)
      IF(LSW1.EQ.2) WRITE(5,39)
 42   FORMAT(// 'UNIT LENGTH :          1 METRE'/ 'UNIT AREA   : 1,000,'
     1'000 SQ.METRES'/ 'UNIT VOLUME : 1,000,000 CU.METRES')
 39   FORMAT(// 'UNIT LENGTH : 1 METRE'/ 'UNIT AREA   : 1 SQ.METRE'/
     1 'UNIT VOLUME : 1 CU.METRE')
      WRITE(5,47) TMBAS
 47   FORMAT(//'UNIT OF TIME FOR DELTA AND BOUNDARY CONDITIONS IS'
     1,2A4)
      WRITE(5,27)
 27   FORMAT(/// 'VALUES OF EXTERNAL SWITCHES'/)
      WRITE(5,28)LSW7,LSW8,LSW9
 28   FORMAT(10X,'LSW7= ',I2/10X,'LSW8= ',I2/10X,'LSW9= ',I2)
      WRITE(5,2)
 2    FORMAT(////' INITIAL WATERTABLE ELEVATIONS'/)
```

```
      WRITE(5,3) T
3     FORMAT(//'      T = ',F10.4//)
      DO 4 K=1,NT
      IF(K.LE.NSCONF) WRITE(5,41) K,H(K),HCONF(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,40)  K,H(K)
      IF(K.GT.NCONF) WRITE(5,45) K,H(K)
41    FORMAT(9H NODE NO.,I4,5X,17HHZERO SEMICONF = ,F8.4,5X,5HH' = ,F8.4
     1)
40    FORMAT(9H NODE NO.,I4,5X,17HHZERO CONFINED = ,F8.4)
45    FORMAT(9H NODE NO.,I4,5X,17HHZERO UNCONF   = ,F8.4)
4     CONTINUE
      DO 90 L=1,LIST
      IF(LSW8.NE.3) GO TO 6
      DO 5 K=1,NN
      TOTRE(K)=0.0
      TOTSQ(K)=0.0
      TOTS(K)=0.0
      TOTFC(K)=0.0
      TOTDQ(K)=0.0
      IF(K.GT.NSCONF) GO TO 5
      TOTSC(K)=0.0
      TOTQS(K)=0.0
5     CONTINUE
      TTRE=0.0
      TTSQ=0.0
      TTS=0.0
      TTSC=0.0
      TTFC=0.0
      TTQS=0.0
      TTDQ=0.0
6     DO 80 M=1,MAJOR
      IF(LSW8.NE.2) GO TO 8
      DO 7 K=1,NN
      TOTRE(K)=0.0
      TOTSQ(K)=0.0
      TOTS(K)=0.0
      TOTFC(K)=0.0
      TOTDQ(K)=0.0
      IF(K.GT.NSCONF) GO TO 7
      TOTSC(K)=0.0
      TOTQS(K)=0.0
7     CONTINUE
      TTRE=0.0
      TTSQ=0.0
      TTS=0.0
      TTSC=0.0
      TTFC=0.0
      TTQS=0.0
      TTDQ=0.0
8     DO 70 JT=1,MINOR
      IF(LSW8.NE.1) GO TO 10
      DO 9 K=1,NN
      TOTRE(K)=0.0
      TOTSQ(K)=0.0
      TOTS(K)=0.0
      TOTFC(K)=0.0
      TOTDQ(K)=0.0
      IF(K.GT.NSCONF) GO TO 9
```

```
      TOTSC(K)=0.0
      TOTQS(K)=0.0
9     CONTINUE
      TTRE=0.0
      TTSQ=0.0
      TTS=0.0
      TTSC=0.0
      TTFC=0.0
      TTQS=0.0
      TTDQ=0.0
10    READ(2) (RECH(K),TOTALQ(K),S(K),FLWCON(K),DRECH(K),K=1,NN)
      READ(2) (H(K),K=1,TNN),T,ITRNO,DELTA,SUBITR
      IF(NSCONF.GT.0) READ(2) (HCONF(K),QSEEP(K),SC(K),K=1,NSCONF)
      DO 11 K=1,NN
      TOTRE(K)=TOTRE(K)+RECH(K)*DELTA
      TOTSQ(K)=TOTSQ(K)+TOTALQ(K)*DELTA
      TOTS(K)=TOTS(K)+S(K)*DELTA
      TOTFC(K)=TOTFC(K)+FLWCON(K)*DELTA
      TOTDQ(K)=TOTDQ(K)+DRECH(K)*DELTA
      IF(K.GT.NSCONF) GO TO 11
      TOTSC(K)=TOTSC(K)+SC(K)*DELTA
      TOTQS(K)=TOTQS(K)+QSEEP(K)*DELTA
11    CONTINUE
      DO 12 K=1,NN
      TTRE=TTRE+RECH(K)*DELTA
      TTSQ=TTSQ+TOTALQ(K)*DELTA
      TTS=TTS+S(K)*DELTA
      TTFC=TTFC+FLWCON(K)*DELTA
      TTDQ=TTDQ+DRECH(K)*DELTA
      IF(K.GT.NSCONF) GO TO 12
      TTSC=TTSC+SC(K)*DELTA
      TTQS=TTQS+QSEEP(K)*DELTA
12    CONTINUE
      WRITE(5,13)T,ITRNO,SUBITR
13    FORMAT(/////5H T = ,F10.4,8X,20HNO. OF ITERATIONS = ,I3,
     18X,23HNO. OF SUBITERATIONS = ,I3/)
      IF(LSW7.NE.1) GO TO 15
      DO 38 K=1,NT
      IF(K.LE.NSCONF) WRITE(5,37) K,H(K),HCONF(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,36) K,H(K)
      IF(K.GT.NCONF) WRITE(5,35) K,H(K)
37    FORMAT(9H NODE NO.,I4,5X,13HH SEMICONF = ,F8.4,5X,5HH' = ,F8.4)
36    FORMAT(9H NODE NO.,I4,5X,13HH CONFINED = ,F8.4)
35    FORMAT(9H NODE NO.,I4,5X,13HH UNCONF   = ,F8.4)
38    CONTINUE
15    IF(LSW8.NE.1) GO TO 70
      WRITE(5,16)
16    FORMAT(//24H WATERBALANCE COMPONENTS)
      WRITE(5,17)
17    FORMAT(//(' NODE',4X,'RECHARGE',4X,'CHANGE IN STORAGE',4X,
     1'SEEPAGE FLOW',4X,'CHANGE IN STORAGE',4X,'TOTAL SUBSURFACE FLOW'
     2,4X,'PUMP FLOW',4X,'DRAINAGE FLOW'/24X,'IN TOPLAYER',27X,
     3'IN AQUIFER'/30X,'(ALL COMPONENTS ARE CALCULATED FOR THE LAST '
     4'TIME STEP)'/))
      DUM=0.0
      DO 18 K=1,NN
      IF(K.LE.NSCONF) WRITE(5,19) K,TOTRE(K),TOTSC(K),TOTQS(K),TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
```

```
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,19) K,DUM,DUM,DUM,TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
      IF(K.GT.NCONF) WRITE(5,19) K,TOTRE(K),DUM,DUM,TOTS(K),TOTSQ(K),
     1TOTFC(K),TOTDQ(K)
19    FORMAT(I5,F12.4,F21.4,F16.4,F21.4,F25.4,F13.4,F17.4)
18    CONTINUE
      WRITE(5,20)
20    FORMAT(//40H WATERBALANCE COMPONENTS FOR WHOLE BASIN/)
      WRITE(5,21) TTRE,TTSC,TTQS,TTS,TTSQ,TTFC,TTDQ
21    FORMAT(F17.4,F21.4,F16.4,F21.4,F25.4,F13.4,F17.4)
   70 CONTINUE
      IF(LSW7.NE.2) GO TO 22
      DO 34 K=1,NT
      IF(K.LE.NSCONF) WRITE(5,37) K,H(K),HCONF(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,36) K,H(K)
      IF(K.GT.NCONF) WRITE(5,35) K,H(K)
34    CONTINUE
22    IF(LSW8.NE.2) GO TO 80
      WRITE(5,16)
      WRITE(5,23) MINOR
23    FORMAT(//(' NODE',4X,'RECHARGE',4X,'CHANGE IN STORAGE',4X,'SEEP'
     1'AGE FLOW',4X,'CHANGE IN STORAGE',4X,'TOTAL SUBSURFACE FLOW',4X,
     1'PUMP FLOW',4X,'DRAINAGE FLOW'/24X,'IN TOPLAYER',27X, 'IN '
     2'AQUIFER'/30X,'(ALL COMPONENTS ARE CALCULATED FOR THE LAST',
     3I3,' TIMESTEPS)'/))
      DO 24 K=1,NN
      IF(K.LE.NSCONF) WRITE(5,19) K,TOTRE(K),TOTSC(K),TOTQS(K),TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,19) K,DUM,DUM,DUM,TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
      IF(K.GT.NCONF) WRITE(5,19) K,TOTRE(K),DUM,DUM,TOTS(K),TOTSQ(K),T
     1OTFC(K),TOTDQ(K)
24    CONTINUE
      WRITE(5,20)
      WRITE(5,21) TTRE,TTSC,TTQS,TTS,TTSQ,TTFC,TTDQ
   80 CONTINUE
      IF(LSW7.NE.3) GO TO 25
      DO 33 K=1,NT
      IF(K.LE.NSCONF) WRITE(5,37) K,H(K),HCONF(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,36) K,H(K)
      IF (K.GT.NCONF) WRITE(5,35) K,H(K)
33    CONTINUE
25    IF(LSW8.NE.3) GO TO 90
      WRITE(5,16)
      IYEAR=MAJOR*MINOR
      WRITE(5,23) IYEAR
      DO 26 K=1,NN
      IF(K.LE.NSCONF) WRITE(5,19) K,TOTRE(K),TOTSC(K),TOTQS(K),TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
      IF(K.GT.NSCONF.AND.K.LE.NCONF) WRITE(5,19) K,DUM,DUM,DUM,TOTS(K)
     1,TOTSQ(K),TOTFC(K),TOTDQ(K)
      IF(K.GT.NCONF) WRITE(5,19) K,TOTRE(K),DUM,DUM,TOTS(K),TOTSQ(K),T
     1OTFC(K),TOTDQ(K)
26    CONTINUE
      WRITE(5,20)
      WRITE(5,21) TTRE,TTSC,TTQS,TTS,TTSQ,TTFC,TTDQ
90    CONTINUE
      END
```

```
*GW4.CSS TO RUN THE PLOTTING MODULE GWMOD4.FTN
DE GWMOD4.PRT
RUN GWMOD4.FTN
PRI GWMOD4.PRT,DEV=EPR:
$EXIT


$TITL GWMOD4.FTN
C GROUNDWATER MODEL PART 4 ; PLOTTING OF RESULTS
      INTEGER TNN
C
C
C
      PARAMETER (NN=5,TNN=15)
      PARAMETER (IMAJOR=12)
C
      DIMENSION A(104),X(11),B(TNN,IMAJOR),H(TNN,IMAJOR),SL( NN),TI(100)
      DIMENSION TMBAS(2),TITLE(40)
C
      OPEN(UNIT=8,FILE='GWMOD4.DAT',STATUS='OLD')
      OPEN(UNIT=5,FILE='GWMOD4.PRT',STATUS='NEW')
      OPEN(UNIT=2,FILE='GWMOD2.BIN',STATUS='OLD')
C
      READ (8,1) BLNK,HI,HMINUS,HX,ASTRSK,HY,PLUS,LSW9,LSW10,LSW11
1     FORMAT (7A4,3I2)
      READ(2) LIST,MAJOR,MINOR,DUM,T,IDUM
      READ(2)(H(K,1),K=1,TNN),NSCONF,(SL(K),K=1,NN),IDUM,TITLE,TMBAS
      IF(NSCONF.GT.0) READ(2)
      IF(LSW10.EQ.2) READ(8,14) MINOR,LIST
14    FORMAT(3I4)
      WRITE(5,33)
33    FORMAT('PLOT'////)
      WRITE(5,29) TITLE
29    FORMAT('GROUNDWATER MODEL FOR',20A4/20A4)
      WRITE(5,31)
31    FORMAT(///'VALUES OF EXTERNAL SWITCHES'/)
      WRITE(5,32) LSW9,LSW10,LSW11
32    FORMAT(10X,8HLSW9   = ,I2/10X,8HLSW10 = ,I2/10X,8HLSW11 = ,I2)
      IF(LSW10.EQ.1) GO TO 35
      WRITE(5,34) MINOR,MAJOR,LIST
34    FORMAT(10X,'MINOR = ',I4/10X,'MAJOR = ',I4/10X,'LIST = ',I4)
35    WRITE(5,30)
30    FORMAT(1H1)
      NO=NN+1
      IF(LSW9.EQ.1) NT=NN
      IF(LSW9.EQ.2) NT=TNN
      DO 90 L1=1,LIST
      DO 80 M=1,MAJOR
      DO 70 JT=1,MINOR
      READ(2)
      IF(JT.GE.MINOR) GO TO 70
      READ(2)
      IF(NSCONF.GT.0) READ(2)
70    CONTINUE
      READ(2) (B(K,M),K=1,TNN),TI(M)
      IF(NSCONF.GT.0) READ(2)
80    CONTINUE
      DO 3 M=1,MAJOR
```

```
      IF(LSW10.EQ.1) READ(8,2) (H(K,M),K=1,NN)
2     FORMAT(8X,9F8.2)
      DO 26 I=1,TNN
      IF(LSW10.EQ.1.AND.I.LE.NN) GO TO 26
      H(I,M)=B(I,M)
26    CONTINUE
3     CONTINUE
      DO 24 II=1,NT
      SMALL=B(II,1)
      HSMALL=H(II,1)
      BIG=B(II,1)
      HBIG=H(II,1)
      DO 4 J=1,MAJOR
      IF(B(II,J).LT.SMALL) SMALL=B(II,J)
      IF(H(II,J).LT.HSMALL) HSMALL=H(II,J)
      IF(SMALL.GT.HSMALL) SMALL=HSMALL
      IF(B(II,J).GT.BIG) BIG=B(II,J)
      IF(H(II,J).GT.HBIG) HBIG=H(II,J)
4     IF(BIG.LT.HBIG) BIG=HBIG
      SCALE=1.0
      IF(BIG-SMALL-1.0.GT.8.0) GO TO 5
      GOTO 8
5     IF(BIG-SMALL-2.0.GT.16.0) GOTO 6
      SCALE=2.0
      GOTO 8
6     IF(BIG-SMALL-4.0.GT.32.0) GOTO 7
      SCALE=4.0
      GOTO 8
7     SCALE=8.0
8     IY=SMALL
      Y=IY
      XINC=-1.0*SCALE
      DO 9 K=1,11
      X(K)=Y+XINC
9     XINC=XINC+SCALE
      IF(LSW10.EQ.1) WRITE(5,10) (X(K),K=1,11)
      IF(LSW10.EQ.2) WRITE(5,28) (X(K),K=1,11)
10    FORMAT(1H+,2X,11(F5.0,5X),17HCOMPT(*)  HIST(+))
28    FORMAT(1H+,2X,11(F5.0,5X),8HCOMPT(*))
      DO 11 I=1,10
      K=I*10-9
      A(K)=HI
      DO 11 J=1,9
      L=J+K
11    A(L)=HMINUS
      A(101)=HI
      DO 12 I=102,104
12    A(I)=BLNK
      WRITE(5,13) A
13    FORMAT(7X,104A1)
      DO 15 K=1,104
15    A(K)=BLNK
      XX=Y-SCALE
      DO 21 J=1,MAJOR
      ITT=TI(J)
      I=(B(II,J)-XX)/(SCALE/10.0)+1.5
      IH=(H(II,J)-XX)/(SCALE/10.0)+1.5
      VAL=B(II,J)
```

```
      HVAL=H(II,J)
      IF(LSW11.EQ.2) GO TO 36
      IF(II.GT.NN) GO TO 36
      B(II,J)=B(II,J)-H(II,J)
36    IF(IH.LE.0.OR.IH.GT.104) GO TO 20
      A(IH)=PLUS
16    IF(I.LE.0.OR.I.GT.104) GOTO 19
      A(I)=ASTRSK
17    IF(LSW10.EQ.1) WRITE(5,18) ITT,A,VAL,HVAL
      IF(LSW10.EQ.2) WRITE(5,18)ITT,A,VAL
18    FORMAT(I6,1X,104A1,2(F9.2))
      A(I)=BLNK
      A(IH)=BLNK
      GO TO 21
19    I=102
      A(I)=HX
      GO TO 17
20    IH=104
      A(IH)=HY
      GOTO 16
21    CONTINUE
      IF(II.LE.NN) WRITE(5,22) II,TMBAS,SL(II)
      IF(II.GT.NN) WRITE(5,27) II,TMBAS
22    FORMAT(// 'NODE NO',I4,13X,'ELEVATIONS (METRES ABOVE SEA LEVEL)
     1VS. TIME' 2A4' SURFACE ELEVATION =  'F4.0)
27    FORMAT(//'NODE NO',I4,26X,' ELEVATIONS(METRES ABOVE SEA LEVEL)
     1VS. TIME '2A4)
      WRITE (5,23)
23    FORMAT(////)
24    CONTINUE
      IF(LSW11.EQ.2) GO TO 90
      XI=0.0
      XI2=0.0
      DO 37 K=1,NN
      DO 38 J=1,MAJOR
      XI=XI+B(K,J)
38    XI2=XI2+B(K,J)*B(K,J)
37    CONTINUE
      MNN=MAJOR*NN
      DMEAN=XI/MNN
      DSDEV=SQRT((XI2-(XI*XI)/MNN)/(MNN-1))
      MT=MAJOR*LIST
      WRITE(5,41)
41    FORMAT('CALCULATION OF DEVIATIONS BETWEEN CALCULATED AND '
     1'HISTORICAL WATERTABLE ELEVATIONS'//)
      WRITE(5,39) MAJOR,MT
39    FORMAT('THIS CALCULATION IS BASED ON THE DEVIATIONS OVER ALL '
     1'THE INTERNAL NODES AND TAKEN OVER ',I3,' TIME STEPS'/
     2'TOTAL NUMBER OF TIME STEPS IN THE ENTIRE CALCULATION IS ',I3/)
      WRITE(5,40) DMEAN,DSDEV
40    FORMAT(' MEAN                 =  ',F5.2, ' METRE'/' STANDARD '
     1'DEVIATION = ',F5.2,' METRE')
      WRITE(5,25)
25    FORMAT(1H1)
90    CONTINUE
      END
```